Lezione 33 (ore 66,67) - 10 dicembre 2025, 11:30 - 13:30 A13

(4.30) <u>Osservazione</u> (scelta di $h(k)$ nei metodi Runge-Kutta).

Coerentemente con l'intento di eliminare l'onere della determinazione e realizzazione delle funzioni $G_j(t,x)$, la scelta di $h(k)$ nei metodi RK avviene, usualmente, come segue.

Siano: RK il metodo di Runge-Kutta, di ordine p per $h \to 0$, scelto per il calcolo di $x(k+1)$ e RK' un altro metodo di Runge-Kutta, di ordine $p' = p+1$. Allora:

- SCELTA di $h(k)$. Dati $E > 0$ e $\lambda > 0$, per ogni k si sceglie τ piccolo, si calcolano:

  (1) XX = un passo di RK a partire da $(x(k),t(k))$, di lunghezza τ
  (2) XX' = un passo di RK' a partire da $(x(k),t(k))$, di lunghezza τ

  si pone:

  $$d(k) = \max \{ \lambda \ , \ \| XX - XX' \| \}$$

  e poi:

  $$h(k) = \min \left\{ \sqrt[p+1]{\frac{E}{d(k)}}\ \tau \ \ , \ tf - t(k) \right\}$$

Questa procedura di scelta si spiega considerando che:

(a) Il metodo RK ha ordine p per $h \to 0$ dunque, posto $C = \dfrac{s^{(p+1)}(0)}{(p+1)!}$:

$$\textit{si stima } s(h) \textit{ con } C\, h^{p+1}$$

(b) Poiché:

$$XX - y(t(k) + \tau) =$$
$$= C\, \tau^{p+1} + z(\tau)\, \tau^{p+1} \ , \quad \text{con } z(\tau) \to 0 \text{ per } \tau \to 0 \quad (\text{RK ha ordine p})$$

$$XX' - y(t(k) + \tau) =$$
$$= C'\, \tau^{p+2} + w(\tau)\tau^{p+2} \ , \quad \text{con } w(\tau) \to 0 \text{ per } \tau \to 0 \quad (\text{RK' ha ordine p+1})$$

allora:

$$\frac{XX - XX'}{\tau^{p+1}} = C + [\ z(\tau) - (C' + w(\tau))\ \tau\ ] \quad \to C \quad \text{per } \tau \to 0$$

e:

$$\textit{scelto } \tau \textit{ piccolo, si stima } C \textit{ con } \frac{XX - XX'}{\tau^{p+1}}$$

(c) Complessivamente:

$$\textit{scelto } \tau \textit{ piccolo, si stima } s(h) \textit{ con } \frac{XX - XX'}{\tau^{p+1}}\ h^{p+1}$$

dunque:

$$\left\| \frac{XX - XX'}{\tau^{p+1}} h^{p+1} \right\| = E \quad \Leftrightarrow \quad h = \sqrt[p+1]{\frac{E}{\| XX - XX' \|}} \; \tau$$

(4.31) Realizzazione in *Scilab* (RK12_pv).

Come esempio di realizzazione, consideriamo il metodo RK che utilizza Eulero esplicito, di ordine 1 per h → 0, per il calcolo di x(k+1) e che sceglie h(k) affiancandolo con il metodo dell'Osservazione (4.23), metodo di Heun di ordine 2 per h → 0. Ne risulta un metodo di ordine 1 per h → 0 e quindi convergente di ordine 1/2 per E → 0.

```
function [T, X, PASSO] = RK12_pv(x0, t0, tf, F, E, LAMBDA, HMIN, TAU)
//
// Integra numericamente, sull'intervallo [t0,tf], il Problema
// di Cauchy in R(n):
//
// x' = F(t,x)
// x(t0) = x0
//
// con il metodo di Eulero esplicito (RK di ordine 1) - a passo
// variabile - affiancato, per la scelta del passo, dal metodo
// RK di ordine 2 definito da c(2) = 1, a(21) = 1 e b(1) = b(2) = 1/2.
//
// x0: condizione iniziale (colonna di n elementi)
// t0: istante iniziale (numero reale)
// tf: istante finale (numero reale)
// F: function che definisce l'equazione differenziale - F(t,x) deve
//    essere una colonna di n numeri reali
// E: valore massimo della stima dell'errore locale (numero reale)
// LAMBDA: numero reale che stabilisce il valore massimo del passo
//         (OPZIONALE - valore predefinito: 1d-5)
// HMIN: valore minimo consentito del passo
//       (OPZIONALE - valore predefinito: (tf - t0) / 1d6)
// TAU: valore del passo per il calcolo delle stime utilizzate
//      nella scelta di h(k) (OPZIONALE - valore predefinito: (tf - t0) / 1d3)
//
// T = [t(0),...,t(N)]: riga contenente gli istanti di integrazione
// X = [x(0),...,x(N)]: matrice n x (N+1) contenente le approssimazioni
// PASSO = [h(0),...,h(N-1)]: riga contenente i passi di integrazione
//
// Valore degli argomenti opzionali
//
if ~exists('LAMBDA','l') then LAMBDA = 1d-5; end;
if ~exists('HMIN','l') then HMIN = (tf - t0) / 1d6; end;
if ~exists('TAU','l') then TAU = (tf - t0) / 1d3; end;
//
// Inizializzazione delle variabili di uscita
//
T(1,1) = t0;
X(:,1) = x0;
PASSO = [];
//
// ciclo principale
```

```
43 //
44 while (T(1,$) < tf), // arresta la costruzione se ha raggiunto tf
45   //
46   // scelta del passo
47   //
48   // XX1 = X(:,$) + F(T(1,$),X(:,$)) * TAU;
49   ST1 = F(T(1,$),X(:,$));
50   ST2 = F( T(1,$) + TAU,X(:,$) + ST1 * TAU );
51   // XX2 = X(:,$) + ( (ST1 + ST2)/2 ) * TAU;
52   //
53   //     XX1 - XX2 = (ST1 - ST2)/2 * TAU
54   //
55   d = max(LAMBDA, norm( ((ST1 - ST2)/2) * TAU ));
56   PASSO(1,$+1) = min(sqrt(E/d) * TAU, tf - T(1,$));
57   //
58   // calcolo approssimazione e nuovo istante di integrazione
59   //
60   X(:,$+1) = X(:,$) + F(T(1,$),X(:,$)) * PASSO(1,$);
61   T(1,$+1) = T(1,$) + PASSO(1,$);
62   //
63   // arresta la costruzione se il passo calcolato risulta troppo
64   // piccolo e non ha raggiunto tf
65   //
66   if (PASSO(1,$) < HMIN) & (T(1,$) < tf) then break; end;
67   //
68 end;
69 //
70 // Verifica se l'integrazione ha raggiunto tf
71 //
72 if T(1,$) < tf then
73    printf("\n\nIntegrazione interrotta a T = %3.2e", T(1,$));
74 end;
75 //
76 endfunction
77 //
78 // Esempio per assegnare valori ai parametri opzionali:
79 //
80 //    [T,X,PASSO] = RK12_pv(x0,t0,tf,F,G,E,HMIN = y);
81 //
82 //         => LAMBDA = valore predefinito, HMIN = y, TAU = valore predefinito
83 //
```

Si osservi che:

- Nella scelta del passo la differenza XX1 - XX2 può essere determinata *senza calcolare* XX1 ed XX2 (righe 48-55). Risulta infatti:

$$\text{XX1} - \text{XX2} = \frac{\text{ST1} - \text{ST2}}{2}\ \text{TAU}$$

- Per la scelta del passo si è utilizzato *lo stesso valore di τ ad ogni iterazione.*

Il file che contiene la procedura, insieme ad un esempio di applicazione all'equazione del pendolo (la stessa dell'Esempio (4.14) della Lezione 30), si può trovare nella pagina web del corso, sezione "altro materiale didattico".